Proudman
Oceanographic
Laboratory

The Operational Storm Surge Model Data Archive

JA Smith
Report No 34 1994



Natural Environment Research Council

PROUDMAN OCEANOGRAPHIC LABORATORY

Bidston Observatory,
Birkenhead, Merseyside, L43 7RA, U.K.

Telephone: 051 653 8633
Telex 628591 OCEANB G
Telefax 051 653 6269

Director: Dr. B.S. McCartney

Natural Environment Research Council

PROUDMAN OCEANOGRAPHIC LABORATORY

REPORT No. 34

The Operational Storm Surge Model Data Archive

Jane A. Smith

March 1994

DOCUMENT DATA SHEET

AUTHOR SMITH, J.A.	PUBLICATION DATE 1994

TITLE

The Operational Storm Surge Model Data Archive

REFERENCE

Proudman Oceanographic Laboratory, Report No. 34, 34pp

ABSTRACT

This report describes the current operational storm surge model data archive. Residual elevations and currents covering the north west European shelf are archived during the storm surge season (September to April). The surge model and the archived data are described, and examples of data extracted from the archive are presented.

Acknowledgements. I would like to thank staff at the Met. Office for their continuing help and support which ensures the smooth running of the Operational Surge Model and its associated data archives. I would also like to thank Dr. Roger Flather and Dr. Roger Proctor for their help and advice. *This work was funded by MAFF.*

ISSUING ORGANISATION

Proudman Oceanographic Laboratory
Bidston Observatory
Birkenhead, Merseyside L43 7RA
UK

Director: Dr B S McCartney

TELEPHONE 051 653 8633

TELEX 628591 OCEAN BG

TELEFAX 051 653 6269

KEYWORDS

OPERATIONAL STORM SURGE MODEL
RESIDUAL CURRENTS
RESIDUAL ELEVATIONS
DATA COLLECTION
DATA ARCHIVES
NORTHWEST EUROPEAN SHELF
NWEUR

CONTRACT

PROJECT MHT-41-1

PRICE £12

Copies of this report are available from:
The Library, Proudman Oceanographic Laboratory.

CONTENTS

1. INTRODUCTION

This report describes the current operational surge model data archive which covers the period from August 1991 to date. This archive is a continuation of a similar archive relating to the previous operational model - containing data from January 1988 to May 1991. Enquiries regarding accessing data from the archive should be made, in the first instance, to *POL Information and Applications Group.*

A storm surge forecast scheme, based on a model developed and maintained by POL for M.A.F.F., has been in routine use at the Meteorological Office since the autumn of 1978. A review of storm surge modelling is given in Heaps (1983). The present scheme makes use of meteorological data from the Limited Area Model (LAM) which runs operationally at the U.K. Meteorological Office for weather forecasting purposes.

The present version of the storm surge model was implemented in 1991 and runs on the Met. Office's CRAY YMP. The previous surge model which ran from 1982 to 1991 (CSX) had a resolution of 35km and was developed to make optimum use of the Met. Office's then new computer system, a CYBER 205E. A detailed account of this system is given in Proctor & Flather (1983). With the installation of a CRAY YMP at the Met. Office, it was feasible to implement a new model (CS3) which had three times the resolution of the existing one (CSX) as the storage and processing capabilities of the new system greatly exceeded those of the old one. This storm surge model (CS3) runs operationally at the Met. Office, twice a day every day throughout the storm surge season (September to May). It runs shortly after the main atmospheric model runs (0z and 12z). This model is of higher resolution (approximately 12km) than the atmospheric model (~50km) and its coverage extends from 48°N to 63°N and from 12°W to 13°E. Recent developments relating to this model are given in Flather and Smith (1993). The surge model grid can be seen in Figure 1. This model computes the tide and surge motion and produces surge forecasts (up to 36 hours ahead) which are used by the Storm Tide Warning Service (STWS) at Bracknell for flood warning purposes.

The model produces data which are used for performance evaluation analysis, as a check on the correct functioning of the system, and archived for other scientific applications. There exists a similar archive (of which this is a continuation) for data produced by the previous operational model (CSX) which is presently held on different storage media (and hence different access methods need to be employed for data retrieval). This report describes the organisation of the current (CS3) operational model data archive - including the surge model formulation (Chapter 2) and a description of the archive format (Chapter 3).

2. MODEL FORMULATION

Tide and surge motion can be represented in terms of the following equations:

$$\frac{\partial \zeta}{\partial t}+\frac{1}{R\cos\varphi}\left\{\frac{\partial}{\partial \chi}(D\bar{u})+\frac{\partial}{\partial \varphi}(D\bar{v}\cos\varphi)\right\}=0 \qquad (1)$$

$$\frac{\partial \bar{u}}{\partial t}+\frac{\bar{u}}{R\cos\varphi}\frac{\partial \bar{u}}{\partial \chi}+\frac{\bar{v}}{R\cos\varphi}\frac{\partial}{\partial \varphi}(\bar{u}\cos\varphi)-2\omega\sin\varphi.\bar{v}=\frac{-g}{R\cos\varphi}\frac{\partial \zeta}{\partial \chi}-\frac{1}{\rho R\cos\varphi}\frac{\partial p_a}{\partial \psi}+\frac{1}{\rho D}(F_S-F_B)+A_h\nabla^2\bar{u} \qquad (2)$$

$$\frac{\partial \bar{v}}{\partial t}+\frac{\bar{u}}{R\cos\varphi}\frac{\partial \bar{v}}{\partial \chi}+\frac{\bar{v}}{R}\frac{\partial \bar{v}}{\partial \varphi}+\bar{u}^2\tan\varphi+2\omega\sin\varphi.\bar{u}=\frac{-g}{R}\frac{\partial \zeta}{\partial \varphi}-\frac{1}{\rho R}\frac{\partial p_a}{\partial \varphi}+\frac{1}{\rho D}(G_S-G_B)+A_h\nabla^2\bar{v} \qquad (3)$$

Equation notation is defined in the glossary. Equation 1 is the equation of continuity which represents a conservation of volume, i.e. inflow or outflow resulting in a change of surface elevation. Equations 2 and 3 state that force is equal to acceleration, per unit mass, where the forces involved are those experienced due to meteorological effects, i.e. wind and atmospheric pressure, and tides. The hydrostatic effect of atmospheric pressure is to raise or lower the sea surface (approximately +1cm for every 1mb below normal atmospheric pressure, and -1cm for every 1mb above). Wind stress is calculated from the mean wind velocity (nominally 10m above the sea surface) by applying the quadratic stress law:

$$\underline{\tau}_S = C_D \rho_a \underline{W}\,|\underline{W}|$$

where C_D is the drag coefficient - a function of sea surface roughness - which is empirically related to the wind speed as follows (Smith and Banke, 1975):

$$C_D \times 10^3 = 0.63 + 0.066\ W$$

Similarly, a quadratic stress law relates the depth mean current to the bottom stress:

$$\underline{\tau}_B = \rho K \underline{q} |\underline{q}|$$

At a coastline, the boundary condition is:

$$q_n = 0$$

where q_n is the component of depth mean current along the outward directed normal to the boundary.

For an open sea boundary, a "radiation" condition is used:

$$q_n = q_n^M + q_n^T + \frac{c}{h}(\zeta - \zeta^M - \zeta^T) \qquad \text{where } c = \sqrt{gh}$$

This condition allows disturbances generated within the model to propagate outwards across the boundary as free progressive waves.

Equations 1 to 3 are solved using finite difference approximations on an Arakawa "C" grid.

In order to determine variations due to meteorological effects, meteorological data from the Met. Office's Limited Area weather forecasting model (LAM) is used. The grid points of this model are shown in Figure 1 - superimposed on the surge model grid. This model produces atmospheric assimilation data (hindcasts - based on observations) and forecast data used for weather prediction. The LAM has a grid size of approximately 50km and covers the North Atlantic Ocean and western Europe. Hourly values of atmospheric pressure at sea level and wind velocity components for the lowest level of this model (σ=0.997, approximately 25m) are used to drive the storm surge model. These values are converted into sea surface elevations equivalent to hydrostatic pressure and wind stress at 10m above sea level, respectively, and then linearly interpolated onto the surge model grid points.

Storm surge residuals are obtained using data output from two model runs. Firstly the model is run for tide only without meteorological data. The second run includes input of meteorological data. The storm surge residuals are obtained by subtracting the results of the first run from the second. This procedure enables tide-surge interaction - an essential process for the propagation of surges in shallow water - to be accounted for. At the Met. Office, the model runs twice a day every day throughout the winter months (August to May) and produces a forecast up to 36 hours ahead, preceded by a 12 hour hindcast which makes use of assimilated meteorological observations. The hindcast provides the best available initial conditions for each forecast. Initial conditions are set to zero at the beginning of the storm surge season (usually some date in August). Initial data for subsequent forecasts are taken from a previous forecast. Data are saved every hour of the 36 hour forecast allowing some continuity should operations be disrupted. A restart with zero initial conditions is only performed if more than three successive forecasts are missing. The forecast procedure is shown schematically in Figure 2. Model arrays of residual elevations and currents at hourly intervals of the hindcast are routinely saved for subsequent use. These data are archived to 3480 cartridge tapes which are subsequently returned to POL. An account of the procedures involved in running the CSX model operationally (which is also applicable to CS3) is given in Flather et al. (1991).

3. THE OPERATIONAL SURGE MODEL ARCHIVE

3.1. Overview

The operational surge model archive contains model arrays of elevation and $\bar{u}$ and $\bar{v}$ components of hourly residual current covering the entire model domain. As previously described, the operational surge model will run a twelve hour hindcast and thirty six hour forecast under normal circumstances. The array data from the hindcast part of the run are archived since they provide the most reliable model information. These data describe the meteorologically driven components of sea surface elevation and depth mean current which are of use for a number of scientific and other applications.

3.2. Data Organisation and Storage

The operational surge model runs from late August to early May - producing data for eight whole months and two partial months. The quantity of data generated throughout the season is too large to transfer via available computer links. Consequently, they are compressed and written to 3480 cartridge tape as part of the operational system. The compression procedure involves the conversion of the data to two byte integers using a simple transformation and makes use of ζ, $\bar{u}$ and $\bar{v}$ masks (which are used to distinguish land and sea points) to further compress the data by not storing data for model points which correspond to land. This effectively reduces both the amount of tape required to store the data operationally at the Met. Office (two tapes per season) and also facilitates data management and storage at POL. Within the model code, calculations are not performed at every point on the grid as a large proportion represents land. An addressing scheme - "compact left and right" - (described in Proctor and Flather, 1983) eliminates the majority of land point calculations. Data compression using this addressing scheme is implemented within many other applications at POL and the recalled archive data can be output in this format. The alternative output format option is a full grid where land points are defined as -99.90. Binary storage is used for both formats (as ascii is impractical for such large data sets). Utilities exist to extract subsets of data based on either of these two formats. For example, a time series or a sub-grid of the model could be extracted for a required period and output in ASCII format.

Typically, one season's data (in compressed archive format) occupies approximately 600MB of storage. Hence, it is inappropriate to store such large quantities of data on on-line disk. Three alternatives are provided:

Firstly, the preferred storage medium is ANSI standard re-writable optical disk. The storage capacity of such a disk is approximately 640MB (320MB per side). Use of optical disk allows direct access as with on-line disk although data access and retrieval times are significantly slower. One season's data (arrays and port data) fit on one optical disk. The optical disk drive is attached to an HP 9000/735 workstation used at POL to support the operational system.

Secondly, another suitable medium is DAT tape. The capacity of a DAT tape is very large (approximately 2GB) which is more than sufficient for archiving requirements. However, as with all tape based storage media, access to data can only be sequential and, as such, will be time consuming if the desired file is not at the beginning of the tape. DAT tape has been used within the archiving procedure primarily for backup purposes, although it can be an alternative if, for example, the optical disk drive is unavailable.

Finally, archived data have been placed on the POL mass storage system. This system is relatively new and its true operational functionality has largely yet to be assessed. Data placed on this system are accessible over the network or can be transferred via ftp to a user's local disk. The configuration of this system means that the files may have automatically been migrated to near-line tape storage. In this case, there will be a delay after a file request while the system recalls the file back to disk cache.

3.3. Archived Data Retrieval

This section outlines general procedures available to authorised users.
Array archived data for each season are stored on one optical disk in five files each containing data for two months either wholly or partially (i.e. August and September, October and November, ... , April and May).

The arrays for one season span two sides of one optical disk. On side A of the optical disk, the directory **cs3*yy*a** contains archived array files. On side B, the archived array files are contained in the directory of the same name as above. The second directory, **ports**, contains the port table archive files (**FILE*nn*.CS3*yy***) (where *nn* is the file number and *yy* is the year) and additionally, an index to the data contained within them (**INDEX.CS3*yy***). The port table archive data are archived on a weekly basis and contain hindcast and forecast data for locations which correspond to "A" class tide gauge locations (shown in Table 1). These data are used primarily for the continual performance evaluation of the operational surge model and are not described further in this report. However these data are also available through *POL Information and Applications Group.*

The array archive files have the following format:

cs3*yy.mmnn*.arrays
cs3*yy.mmnn*.headers

where *yy* is the year of the season (i.e. 94 for the 1993-94 season) and *mm* is the first month for which data are stored and *nn* is the second month for which data are stored. The file with extension **.headers** contains a header for each entry (every 12 hours) in the data file with the extension **.arrays**.

As described above, files containing arrays are additionally backed up onto DAT. A subset of one season's data can be extracted by logging in to a workstation which has a DAT drive attached. The file can be read using the **dd** command specifying a block size of **800kB** after skipping to the required file using the command **mt fsf** *n* (where *n* is the number of files to skip.

Each file is large (approximately 100MB). It is therefore essential to ensure that there is sufficient disk space available to accommodate the restored files. Once restored, the arrays can be reconstituted into the standard "compact left and right" format (with standard date/time header) or into arrays which cover the full surge model grid (including land points) by using the program **archive.f**. It is necessary to modify the files **archive.fil** and **archive.dat** prior to execution.

The file **archive.fil** contains the input filename being used and the output filename for the recalled data. The other three records remain the same.

The file **archive.dat** contains three records. Record 1 contains the hour, day, month and year of the first data to be extracted. Record 2 is the number of records to be processed. Record three is a flag for specifying which variables are to be output. Data are handled in twelve hour segments and hence the minimum amount of data which can be extracted in one go is twelve hourly sets of arrays. The hour value entered in this file must be 00 or 12. The number of records to be processed is required to be entered on the next line. In the archive, there are two records per day - each record containing twelve hours of data. If data are missing the archive retrieval program will process this number of records regardless of any data that are missing. It is advisable to check the contents of the archive arrays file by interrogating the associated header file prior to executing the program **archive.f** as occasionally there are missing data. The next record is a flag which determines the variables to be stored on recall. There are two options. Firstly, if the flag is set to 0, elevations only are output. This is useful for applications where only elevations are required because the storage space required is reduced by two thirds. Secondly, setting the flag to 1 will enable output of all three components. In the latter case, the disk space required increases. One month's data for all three components of flow will occupy approximately 130MB. The final record in this file is another flag which is used to determine the output format of the recalled arrays. If the flag is set to 1 then the output arrays are written in the "compact left and right" format. If the flag is set to 0 then values at every grid point are output (land points are set to -99.90).

The file **archive.pri** contains the headers corresponding to the actual data read from the archive in order to produce the final output file. If the list is discontinuous then this is indicative of a gap in the archive. A listing of the program **archive.f** and its associated files can be found in the Appendix.

December 1993 was a notable month for surge events. Examples of output from the surge model archive for this month can be seen in Figures 3-6. Figure 3 shows a time series of model and observed residual elevations for Lowestoft. Figures 4 and 5 show surge current vectors and surge elevation distributions during a surge event. A depression with central pressure of 996mb crossed the North Sea from Scotland to Denmark on 9th December with strong westerly winds to the south generating the circulation shown in Figure 4 and a surge of 2.8m on the Dutch coast (Figure 5). Other information can be obtained from archived data. For example, storm surges account for much of the variability in the mean circulation and sea level on timescales of one month and longer. The mean residual circulation for December 1993, calculated from the surge model archive data, is shown in Figure 6. Enquiries regarding accessing data from the archive should be made to *POL Information and Applications Group.*

REFERENCES

FLATHER, R.A., PROCTOR, R. & WOLF, J. 1991. Oceanographic Forecast Models. pp. 15-30 in, Computer Modelling in the Environmental Sciences, (ed. D.G. Farmer & M.J. Rycroft). Oxford: Clarendon Press. 379pp.

FLATHER, R.A. & SMITH J.A. 1993. Recent Progress with Storm Surge Models - Results for January and February 1993. Proceedings of the MAFF Conference of River and Coastal Engineers - Loughborough University 1993.

HEAPS, N.S. 1983. Storm Surges, 1967-1982. Geophysical Journal of the Royal Astronomical Society, 74, 331-376.

PROCTOR, R. & FLATHER R.A. 1983. Routine Storm Surge Forecasting Using Numerical Models: Procedures and Computer Programs for use on the CDC Cyber 205 at the British Meteorological Office. Institute of Oceanographic Sciences, Report, No.167, 171pp.

SMITH, S.D. & BANKE, E. 1975. Variation of the Sea Surface Drag Coefficient with Wind Speed. Quarterly Journal of the Royal Meterological Society, 101, 665-673.

GLOSSARY

χ,φ	longitude and latitude
t	time
ζ	elevation of the sea surface
$\bar{u},\bar{v}$	east and north components of depth mean current, $\boldsymbol{q}$
D	total depth of water ($h+\zeta$)
h	still water depth
R	radius of the Earth
ω	angular speed of rotation of the Earth
g	acceleration due to gravity
ρ	density of sea water (1025 kgm^{-3})
p_a	surface atmospheric pressure
F_s,G_s	components of wind stress, τ_S, on the sea surface
F_B,G_B	components of bottom stress, τ_B
ρ_a	density of air (1.25 kgm^{-3})
W	wind speed (ms^{-1})
K	friction parameter (0.0025)
^{M}	input of meteorological origin
^{T}	input of tidal origin
σ	the "normalised" vertical coordinate system used in the LAM
A_h	horizontal eddy viscosity parameter

PORTNO	NAME	COUNTRY	LATITUDE	LONGITUDE
38	ABERDEEN	SCOTLAND	57deg 8′ 38.9″N	2deg 4′ 43.2″W
60	AVONMOUTH	ENGLAND	51deg 30′ 36.9″N	2deg 42′ 50.7″W
923	BARMOUTH	WALES	52deg 43′ 8.4″N	4deg 2′ 37.8″W
183	CROMER	ENGLAND	52deg 56′ 1.9″N	1deg 18′ 12.5″E
2	DEVONPORT	ENGLAND	50deg 22′ 37.9″N	4deg 11′ 0.8″W
12	DOVER	ENGLAND	51deg 6′ 59.7″N	1deg 19′ 5.4″E
204	FELIXSTOWE	ENGLAND	51deg 57′ 22.8″N	1deg 21′ 0.0″E
55	FISHGUARD	WALES	52deg 0′ 46.2″N	4deg 58′ 57.5″W
50	HEYSHAM	ENGLAND	54deg 2′ 0.3″N	2deg 54′ 41.7″W
3	HINKLEY	ENGLAND	51deg 13′ 0.0″N	3deg 8′ 0.0″W
54	HOLYHEAD	WALES	53deg 18′ 49.3″N	4deg 37′ 9.4″W
61	ILFRACOMBE	ENGLAND	51deg 12′ 39.0″N	4deg 6′ 36.3″W
26	IMMINGHAM	ENGLAND	53deg 37′ 58.9″N	0deg 11′ 13.0″W
919	I.O.M. PORT ERIN	ISLE OF MAN	54deg 5′ 7.0″N	4deg 45′ 55.0″W
202	ISLAY	SCOTLAND	55deg 38′ 0.0″N	6deg 11′ 0.0″W
74	JERSEY (ST. HELIER)	JERSEY	49deg 11′ 0.0″N	2deg 7′ 0.0″W
918	KINLOCHBERVIE	SCOTLAND	58deg 27′ 22.6″N	5deg 2′ 58.2″W
34	LEITH	SCOTLAND	55deg 59′ 23.3″N	3deg 10′ 48.9″W
41	LERWICK	SCOTLAND	60deg 9′ 0.0″N	1deg 8′ 0.0″W
52	LIVERPOOL, PRINCES PIER	ENGLAND	53deg 24′ 30.0″N	2deg 59′ 54.0″W
7	LLANDUDNO	WALES	53deg 20′ 0.0″N	3deg 50′ 0.0″W
24	LOWESTOFT	ENGLAND	52deg 28′ 20.9″N	1deg 45′ 6.4″E
56	MILFORD HAVEN	WALES	51deg 42′ 21.5″N	5deg 3′ 2.1″W
49	MILLPORT	SCOTLAND	55deg 44′ 58.2″N	4deg 54′ 17.9″W
932	MUMBLES	WALES	51deg 34′ 0.0″N	3deg 58′ 0.0W
1	NEWLYN	ENGLAND	50deg 6′ 8.7″N	5deg 32′ 30.0″W
11	NEWHAVEN	ENGLAND	50deg 47′ 0.0″N	0deg 4′ 0.0″E
57	NEWPORT (GWENT)	WALES	51deg 33′ 0.0″N	2deg 59′ 0.0″W
32	NORTH SHIELDS	ENGLAND	55deg 0′ 26.1″N	1deg 26′ 17.9″W
63	PORTPATRICK	SCOTLAND	54deg 50′ 32.7″N	5deg 7′ 8.0″W
8	PORTSMOUTH	ENGLAND	50deg 48′ 1.3″N	1deg 6′ 36.6″W
15	SHEERNESS	ENGLAND	51deg 26′ 42.4″N	0deg 44′ 41.9″E
231	ST. MARYS, (SCILLY ISLES)	ENGLAND	49deg 55′ 0.0″N	6deg 19′ 0.0″W
42	STORNOWAY	HEBRIDES	58deg 12′ 28.6″N	6deg 23′ 17.5″W
223	TOBERMORY	SCOTLAND	56deg 37′ 23.3″N	6deg 3′ 46.1″W
43	ULLAPOOL	SCOTLAND	57deg 53′ 44.0″N	5deg 9′ 26.9″W
991	WEYMOUTH	ENGLAND	50deg 34′ 0.0″N	2deg 26′ 0.0″W
174	WHITBY	ENGLAND	54deg 29′ 23.7″N	0deg 36′ 45.4″W
35	WICK	SCOTLAND	58deg 26′ 28.5″N	3deg 5′ 5.7″W
217	WORKINGTON	ENGLAND	54deg 39′ 0.0″N	3deg 34′ 0.0″W

TABLE 1. "A" Class Tide Gauge Locations



FIGURE 1. The Operational Surge Model Grid with Atmospheric Model (LAM) Grid Points



FIGURE 2. Schematic Diagram Showing Surge Forecast Procedure



FIGURE 3. Model (————) and Observed (+ + + +) Residual Elevations at Lowestoft



FIGURE 4. Model Residual Currents for 12z 09/12/93



FIGURE 5. Model Residual Elevations at 12z 09/12/93



FIGURE 6. Monthly Mean Residual Currents for December 1993

APPENDIX

Archive Data Extraction Program and Associated files

```
archive.fil
===========

archive.dat                             ! Input data
archive.pri                             ! Print out
setupn7c.uda                            ! Model setup data
cs394.1201.arrays                       ! Input file (from archive)
dec93.uda                               ! Output data
```

```
archive.dat
===========

00,01,12,1993     ! hour,day,month,year   NB hour must be 00 or 12
62                ! number of records to be processed (2 per day)
0                 ! flag for output (1 = z only; z, u & v otherwise)
1                 ! flag for output (1 = comressed; 0 = full grid)
```

```
C     ******************************************************************
C                                                                      *
C     ARCHIVE - Recall archive surge model data from file and convert  *
C     =======   into standard (compact left + right or fully expanded) *
C               format.                                                *
C                                                                      *
C     For CS3 model archive data from Operational Surge Model.         *
C                                                                      *
C     Modifications:                                                   *
C     -------------                                                    *
C                                                                      *
C     Writes header for each array set                       15/01/92  *
C     This one for new archive format                        13/08/92  *
C     Version for UNIX                                       13/05/92  *
C     Error in transformation choice logic removed           18/06/93  *
C     Write header as i*4 instead of i*2                     27/01/94  *
C     Change header hour count to be consistent                        *
C                             with model format              27/01/94  *
C                                                                      *
C     Jane Smith                                                       *
C     ==========                                                       *
C                                                                      *
C     ******************************************************************
      PARAMETER (NTOT=15500,LIM=500,NTOTX=20250,NTOT3=NTOT*3)
C
C     NTOT  = compact arrays
C     LIM   = limits and open boundaries
C     NTOTX = expanded arrays
C
      INTEGER*2 IARC(NTOT3),IHEADH(10),IHEADS(10)
      INTEGER   HEADER(5)
      DIMENSION H(NTOT),CSP(LIM),CRP(LIM)
      DIMENSION Z(NTOT),U(NTOT),V(NTOT)
      DIMENSION ZZ(NTOTX),UU(NTOTX),VV(NTOTX)
      DIMENSION IZOB(LIM),LABZ(LIM),LBAD(LIM),IUOB(LIM),IVOB(LIM)
      DIMENSION IZ1(LIM),IZ2(LIM),IU1(LIM),IU2(LIM),IV1(LIM),IV2(LIM)
      DIMENSION IUM(NTOT),IVM(NTOT),IZM(NTOT)
      COMMON/COM2/NPRI,INT(LIM)
      COMMON/COM4/NCC,NRR,ITOT
      COMMON/COM9/NTRNS(LIM),NTRNT(LIM),NSUM(LIM)
      CHARACTER*35 DATA,PRINT,INPUT,OUTPUT,SETUP
C
C     file unit numbers
C
      NARC = 10
      NPRI = 11
      NRD  = 12
      NOUT = 20
      NFIL = 15
      NRA  = 16
C
C     open files
C
      OPEN(NFIL,FILE='archive.fil',STATUS='OLD',FORM='FORMATTED')
      READ(NFIL,111)DATA,PRINT,SETUP,INPUT,OUTPUT
 111  FORMAT(A35)
      OPEN(NRD,file=data,status='old',form='formatted')
      OPEN(NPRI,file=print,status='unknown',form='formatted')
      OPEN(NRA,file=setup,status='old',form='unformatted')
      OPEN(NARC,file=input,status='old',form='unformatted')
      OPEN(NOUT,file=output,status='unknown',form='unformatted')
C
C     read sea model control data from disc
C
      READ(NRA) NRR,NCC,ITOT,IINZ
      READ(NRA) (NTRNS(I),I=1,NRR)
```

```
      READ(NRA) (NTRNT(I),I=1,NRR)
      READ(NRA) (NSUM(I),I=1,NRR)
C
C     NTRANS,NTRNT,NSUM transformation matrices for compact addressing
C
      READ(NRA) ILIZ
C
C     ILIZ=Number of Z-limits
C
      READ(NRA) (IZ1(J),J=1,ILIZ)
      READ(NRA) (IZ2(J),J=1,ILIZ)
      READ(NRA) ILIU
C
C     ILIU=Number of U-limits
C
      READ(NRA) (IU1(J),J=1,ILIU)
      READ(NRA) (IU2(J),J=1,ILIU)
      READ(NRA) ILIV
C
C     ILIV=Number of V-limits
C
      READ(NRA) (IV1(J),J=1,ILIV)
      READ(NRA) (IV2(J),J=1,ILIV)
      READ(NRA) IOBZ
C
C     IOBZ=Number of open boundary Z-points
C
      READ(NRA) (IZOB(J),J=1,IOBZ)
      READ(NRA) (LABZ(J),J=1,IOBZ)
      READ(NRA) (LBAD(J),J=1,IOBZ)
      READ(NRA) DX,DY,ILAT,IMEU,IMEV,IOBU,IOBV
C
C     DX   Grid increment in X-direction..East
C     DY   Grid increment in Y-direction..North
C     ILAT=Number of latitudes covered
C     IMEU=Number of U-met data points
C     IMEV=Number of V-met data points
C     IOBU=Number of open boundary U-points
C     IOBV=Number of open boundary V-points
C
      READ(NRA) (H(I),I=1,ITOT)
      IF(IOBU.NE.0) READ(NRA) (IUOB(J),J=1,IOBU)
      IF(IOBV.NE.0) READ(NRA) (IVOB(J),J=1,IOBV)
      READ(NRA) (CRP(MLAT),MLAT=1,ILAT)
C
C     compute z,u,v masks
C
      DO 788 I=1,ITOT
      IUM(I)=0
      IVM(I)=0
      IZM(I)=0
  788 CONTINUE
      DO 789 J=1,ILIU
      I1=IU1(J)
      I2=IU2(J)
      DO 789 I=I1,I2
      IUM(I)=1
  789 CONTINUE
      DO 787 J=1,IOBU
      I=IUOB(J)
      IUM(I)=1
  787 CONTINUE
      DO 790 J=1,ILIV
      I1=IV1(J)
      I2=IV2(J)
      DO 790 I=I1,I2
```

```
      IVM(I)=1
  790 CONTINUE
      DO 773 J=1,IOBV
      I=IVOB(J)
      IVM(I)=1
  773 CONTINUE
      DO 791 J=1,ILIZ
      I1=IZ1(J)
      I2=IZ2(J)
      DO 791 I=I1,I2
      IZM(I)=1
  791 CONTINUE
      DO 771 J=1,IOBZ
      I=IZOB(J)
      IZM(I)=1
  771 CONTINUE
C
C     read control data
C
      READ(NRD,*)IHHS,IDDS,IMMS,IYYS
      READ(NRD,*)NSETS
      READ(NRD,*)IZFLAG
      READ(NRD,*)IFORM
      WRITE(NPRI,*)IHHS,IDDS,IMMS,IYYS
      WRITE(NPRI,*)NSETS
      WRITE(NPRI,*)IZFLAG
      WRITE(NPRI,*)IFORM
C
C     check data is within file limits
C
      READ(NARC)(IHEADH(I),I=1,9)
      NVAL=IHEADH(7)+IHEADH(8)+IHEADH(9)
      DO 15 L=1,IHEADH(6)
 15     READ(NARC)(IARC(I),I=1,NVAL)
      IDYFS=IHEADH(2)
      IMNFS=IHEADH(3)
      IYRFS=IHEADH(4)
      CALL VDAY(IDYFS,IMNFS,IYRFS,IDYNOFS)
      CALL VDAY(IDDS,IMMS,IYYS,IDYNOS)
      WRITE(NPRI,*)'IDYNOS,IDYNOFS',IDYNOS,IDYNOFS
      IF (IDYNOS.LT.IDYNOFS) THEN
        IFLAG=1
        GOTO 999
      END IF
      NSKIP=((IDYNOS-IDYNOFS)*2)+IHHS/12
      WRITE(NPRI,*)'NSKIP=',NSKIP
      REWIND (NARC)
C
C     skip data to start of requested period
C
      DO 25 K=1,NSKIP
         READ(NARC)(IHEADH(I),I=1,9)
         NVAL=IHEADH(7)+IHEADH(8)+IHEADH(9)
         DO 25 L=1,IHEADH(6)
 25         READ(NARC)(IARC(I),I=1,NVAL)
C
C     now process required data
C
      DO 70 KOUNT=1,NSETS
C
C     read header
C
      READ(NARC,END=999)(IHEADH(I),I=1,9)
c
c     read (iheadh(6)) hours of data
c
```

```
      KOUNTER=IHEADH(1)
      DO 60 L=1,IHEADH(6)
      READ(NARC)(IARC(I),I=1,NVAL)
C
C     The next bit checks that data is not from 1991-2 season as the
C     transformation is slightly different. It was subsequently
C     changed due to errors in high surge levels being clipped.
C
      IF(IYYS.GT.1992.OR.(IYYS.EQ.1992.AND.IDYNOS.GT.200)) THEN
C
C     expand onto compressed grid using z,u and v masks
C
      J=0
      DO 30 KK=1,ITOT
         IF(IZM(KK).EQ.0) Z(KK)=0.00000000
         IF(IZM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           Z(KK)=(DUM+6553.6)/6553.6
         END IF
 30   CONTINUE
      DO 40 KK=1,ITOT
         IF(IUM(KK).EQ.0) U(KK)=0.0000000
         IF(IUM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           U(KK)=(DUM+6553.6)/6553.6
         END IF
 40   CONTINUE
      DO 50 KK=1,ITOT
         IF(IVM(KK).EQ.0) V(KK)=0.0000000
         IF(IVM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           V(KK)=(DUM+6553.6)/6553.6
         END IF
 50   CONTINUE
      ELSE
C
C     expand onto compressed grid using z,u and v masks (1991-2 season)
C
      J=0
      DO 31 KK=1,ITOT
         IF(IZM(KK).EQ.0) Z(KK)=0.00000000
         IF(IZM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           Z(KK)=(DUM-25000.)/6000.
         END IF
 31   CONTINUE
      DO 41 KK=1,ITOT
         IF(IUM(KK).EQ.0) U(KK)=0.0000000
         IF(IUM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           U(KK)=(DUM-25000.)/6000.
         END IF
 41   CONTINUE
      DO 51 KK=1,ITOT
         IF(IVM(KK).EQ.0) V(KK)=0.0000000
         IF(IVM(KK).EQ.1) THEN
           J=J+1
           DUM=IARC(J)
           V(KK)=(DUM-25000.)/6000.
         END IF
 51   CONTINUE
```

```
      END IF
C
C     write data out in compressed form
C
      DO 55 KK=1,4
 55   HEADER(KK)=IHEADH(KK)
      HEADER(5)=KOUNTER-HEADER(1)
      WRITE(NOUT)(HEADER(I),I=1,5)
      WRITE(6,*)(HEADER(I),I=1,5)
C
C     compact left and right
C
      IF(IFORM.EQ.1) THEN
      IF(IZFLAG.EQ.1) THEN
      WRITE(NOUT)(Z(I),I=1,ITOT)
      ELSE
      WRITE(NOUT)(Z(I),I=1,ITOT)
      WRITE(NOUT)(U(I),I=1,ITOT)
      WRITE(NOUT)(V(I),I=1,ITOT)
      END IF
      END IF
C
C     full model grid
C
      IF(IFORM.EQ.0) THEN
      IF(IZFLAG.EQ.1) THEN
      CALL EXPAND(Z,ZZ)
      WRITE(NOUT)(ZZ(I),I=1,NTOTX)
      ELSE
      CALL EXPAND(Z,ZZ)
      CALL EXPAND(U,UU)
      CALL EXPAND(V,VV)
      WRITE(NOUT)(ZZ(I),I=1,NTOTX)
      WRITE(NOUT)(UU(I),I=1,NTOTX)
      WRITE(NOUT)(VV(I),I=1,NTOTX)
      END IF
      END IF
      KOUNTER=KOUNTER+1
 60   CONTINUE
C
C     write header data to print file
C
      WRITE(NPRI,2222)(IHEADH(j),j=1,9)
 2222 FORMAT(9I8)
   70 CONTINUE
  999 CONTINUE
C
C     print array to test data is OK
C
Ctest call prntz(z)
C
C     report errors
C
      IF (IFLAG.EQ.1) THEN
       WRITE(6,*)
       WRITE(6,*)'ERROR - selected date out of range'
       END IF
      STOP
      END
C =================================================================
      SUBROUTINE PRNTZ(A)
C =================================================================
C
C     PRNTZ -
C     Print model array in appropriate format
C
```

```
C ==================================================================
      PARAMETER (NTOT=15500,LIM=500)
      DIMENSION A(NTOT),AOUT(LIM)
      COMMON/COM2/NW,INT(LIM)
      COMMON/COM4/NCC,NRR,ITOT
      COMMON/COM9/NTRNS(LIM),NTRNT(LIM),NSUM(LIM)
  110 FORMAT(1H //3X,20I6)
  111 FORMAT(2X,I3,1X,20F6.3)
  112 FORMAT(1H0//////3X,20I6)
      IS=1
      IND=20
      IND1=IND-1
      ISTOP=IS+IND1
      STAR=10000001.
      NCYC=(NCC/IND)+1
      NREM=NCC-(IND*(NCYC-1))
      IF(NREM.EQ.0) NCYC=NCYC-1
      WRITE(NW,110) (INT(J),J=IS,ISTOP)
      DO 99 ICYC=1,NCYC
      IROS=1
      IROE=NRR
      IF(ICYC.EQ.1) GO TO 98
      WRITE(NW,112) (INT(J),J=IS,ISTOP)
   98 CONTINUE
C     *****************************************************************
C     Save paper  CS3 model
C     *****************************************************************
      IF(ICYC.EQ.5.AND.NRR.EQ.135) IROE=120
      IF(ICYC.EQ.6.AND.NRR.EQ.135) IROE=93
      IF(ICYC.EQ.7.AND.NRR.EQ.135) IROS=28
      IF(ICYC.EQ.7.AND.NRR.EQ.135) IROE=88
      IF(ICYC.EQ.8.AND.NRR.EQ.135) IROS=40
      IF(ICYC.EQ.8.AND.NRR.EQ.135) IROE=82
C     *****************************************************************
      DO 90 K=IROS,IROE
      I3=INT(K)
      NSTART=NTRNT(K)+1
      NSTOP=NTRNS(K)
      NSUMK1=NSUM(K)-(NSTOP-NSTART)
      DO 990 I=1,NCC
  990 AOUT(I)=STAR
C     Transfer Kth row of compact data to print array
      II=0
      DO 30 I=NSTART,NSTOP
      AA=A(NSUMK1+II)
      IF(AA.NE.0.) AOUT(I)=AA
      II=II+1
   30 CONTINUE
C     Write page width segment of print array
      WRITE(NW,111) I3,(AOUT(J),J=IS,ISTOP)
   90 CONTINUE
      IS=ISTOP+1
      ISTOP=IS+IND1
      IF(ISTOP.GT.NCC) ISTOP=NCC
   99 CONTINUE
      RETURN
      END
C ==================================================================
      SUBROUTINE VDAY(IDAY,IMNTH,IY,IVDY)
C ==================================================================
C
C     VDAY   -
C     Convert  (day, month, year)  to  (day number, year)
C
C ==================================================================
       IF(IMNTH.EQ.1)  IVDY=IDAY
```

```
       IF(IMNTH.EQ.2)  IVDY=IDAY+31
       IF(IMNTH.EQ.3)  IVDY=IDAY+59
       IF(IMNTH.EQ.4)  IVDY=IDAY+90
       IF(IMNTH.EQ.5)  IVDY=IDAY+120
       IF(IMNTH.EQ.6)  IVDY=IDAY+151
       IF(IMNTH.EQ.7)  IVDY=IDAY+181
       IF(IMNTH.EQ.8)  IVDY=IDAY+212
       IF(IMNTH.EQ.9)  IVDY=IDAY+243
       IF(IMNTH.EQ.10) IVDY=IDAY+273
       IF(IMNTH.EQ.11) IVDY=IDAY+304
       IF(IMNTH.EQ.12) IVDY=IDAY+334
      IYR=IY
        IF(MOD(IYR,4).EQ.0.AND.IMNTH.GT.2)    IVDY=IVDY+1
        IF(MOD(IYR,100).EQ.0.AND.IMNTH.GT.2) IVDY=IVDY-1
        IF(MOD(IYR,400).EQ.0.AND.IMNTH.GT.2) IVDY=IVDY+1
      RETURN
      END
C ==================================================================
      SUBROUTINE EXPAND(Z,ZZ)
C ==================================================================
C
C    EXPAND -
C    Expand "compact left and right" model data to full grid
C
C ==================================================================
      PARAMETER (NTOT=15500,LIM=500,NTOTX=20250,NTOT3=NTOT*3)
      DIMENSION Z(NTOT),U(NTOT),V(NTOT)
      DIMENSION ZZ(NTOTX),UU(NTOTX),VV(NTOTX)
      COMMON/COM4/NCC,NRR,ITOT
      COMMON/COM9/NTRNS(LIM),NTRNT(LIM),NSUM(LIM)
      DO 1 KK=1,NTOTX
      ZZ(KK)=-99.90
      UU(KK)=-99.90
 1    VV(KK)=-99.90
      DO 5 I=1,NRR
       NSTART=NTRNT(I)+1
       NSTOP=NTRNS(I)
       NSUM1=NSUM(I)-(NSTOP-NSTART)
       II=0
       DO 5 J=NSTART,NSTOP
          M=NSUM1+II
          K=(I-1)*NCC+J
          ZZ(K)=Z(M)
          UU(K)=U(M)
          VV(K)=V(M)
          II=II+1
 5        CONTINUE
      RETURN
      END
```